



LexML Brasil
Parte 2 – LexML URN

Versão 1.0 (RC1)
Brasília – Dezembro de 2008

LexML Brasil

Parte 2 – LexML URN

Versão 1.0 (Dezembro/2008)

Parte 2 – LexML URN	2
1. Introdução.....	4
1.1. Objetivo do Nome Uniforme do Recurso.....	5
1.2. Associação do Nome Uniforme ao Documento	5
1.3. Utilização do Nome Uniforme nas Referências	5
1.4. Tipos de URN.....	6
1.5. Princípios-base do Nome Uniforme (URN)	6
1.6. Características Gerais do Sistema de Identificação e Referência a Documentos	7
2. Sintaxe Utilizada neste Documento.....	7
3. Espaço de nomes (namespace)	8
3.1. Internacionalização do espaço de nomes "lex"	8
4. Sintaxe Geral do Nome-Específico Brasileiro	9
4.1. Caracteres Permitidos e Proibidos.....	9
4.2. Caracteres Reservados.....	10
4.3. Casos Especiais	10
4.4. Formato da Data	12
5. Construção do Nome-Específico (NSS-lex-br)	12
5.1. Estrutura do Nome-específico <NSS-lex-br>	12
5.2. Nomes Alternativos (alias)	13
6. Elemento <local>.....	14
7. Elemento <documento>.....	15
8. Elemento <autoridade>	16
8.1. Indicação da Autoridade.....	16
8.2. Pluralidade dos Sujeitos	16
8.3. Modalidade de Indicação do Sujeito	16
9. Elemento <tipo-documento>	18
9.1. Critérios para a Indicação do Tipo do Documento.....	18
9.2. Especificação Adicional do Tipo de Documento	18
9.3. Especificação da Publicação Oficial.....	19
10. Elemento <descriptor>	19

10.1.	Indicação do Descritor.....	20
10.2.	Elemento <identificadores>.....	21
10.3.	Identificador de Componentes de um Documento	23
10.4.	Descritor da Publicação Oficial.....	25
11.	Elemento <fragmento>	25
11.1.	Fragmento da Publicação Oficial	26
12.	Elemento <versao>.....	27
12.1.	As Versões e Visões de um Texto	27
12.2.	Casos Especiais	28
12.3.	Identificação da Versão	28
12.4.	Exemplo - CDC	29
12.5.	Valores <i>Default</i> para os Elementos da Versão	30
13.	Elemento <forma>.....	30
14.	Referências	31
14.1.	Bibliográficas	31
14.2.	Sites	31
15.	Lista de Abreviaturas e Siglas	31
16.	Créditos.....	32
16.1.	Editores (LexML Brasil)	32
16.2.	Grupo de Trabalho LexML (em Dezembro de 2008).....	32
16.3.	Outras Contribuições	33
Anexo A – Gramática da URN Canônica.....		34
Anexo B – Gramática da URN de Referência.....		38

1. Introdução

Os nomes uniformes de recursos - *Uniform Resource Names* (URNs) – foram concebidos para fornecer identificadores unívocos e persistentes para recursos publicados na Internet, independente da sua localização na rede.

Com este documento define-se a identificação de normas, julgados e projetos de normas por meio de um nome uniforme para o ordenamento jurídico brasileiro. Tal nome será utilizado como instrumento para representar as referências (e, mais genericamente, quaisquer tipos de relações) entre os documentos, com a finalidade de tornar disponível um ambiente hipertextual global sobre os recursos de informação, operacional em um ambiente distribuído.

Em especial, este documento descreve as regras para a atribuição de nomes uniformes e delinea os critérios principais para que os serviços de resolução possam associar a um nome uniforme a localização do documento correspondente na rede.

Nos documentos jurídicos são muito freqüentes, e extremamente importantes, as referências a outros documentos; portanto, a possibilidade de poder consultar imediatamente o documento referenciado constitui um auxílio de grande relevância, e, às vezes, indispensável para a plena compreensão do texto.

Os links hipertextuais da web certamente respondem a essa necessidade, mas mostram-se inadequados a um emprego em larga escala para referenciar documentos normativos e legislativos, pois a referência ao recurso relacionado baseia-se, na maioria das vezes, em sua localização física, expressa de modo uniforme por meio de uma URL (*Uniform Resource Locator*), que apresenta problemas já amplamente conhecidos, quais sejam:

- a dificuldade de se conhecer a localização do recurso citado;
- com o passar do tempo, a perda de validade dos links (URL) nas referências;
- a impossibilidade de referenciar recursos ainda não publicados;

Esses problemas tornam a rede de links entre documentos extremamente limitada com relação às suas potencialidades e cada vez menos confiável com o passar do tempo.

Para remediar tais inconvenientes, utiliza-se um sistema de referências baseado na atribuição a cada documento de um nome uniforme (URN: *Uniform Resource Name*), ou identificador unívoco, e um método de resolução (RDS: *Resolver Discovery Service*) capaz de encontrar o objeto correspondente. Essa abordagem está de acordo com as recomendações do IETF (*Internet Engineering Task Force*), definidas por um grupo de trabalho específico (*URN Working Group*), e descritas em vários documentos - dos padrões oficiais (*RFC: Request For Comments*) aos *drafts* - aos quais é, e deverá ser, garantido o alinhamento também no futuro.

1.1. Objetivo do Nome Uniforme do Recurso

A atribuição de um nome uniforme tem como objetivo associar a cada documento jurídico, legislativo ou parte dele um identificador unívoco, em um formato padronizado que dependa apenas das características do próprio documento (metadados) e que, por isso, seja independente da disponibilidade em rede, da sua localização física e das modalidades de acesso.

Tal identificador é utilizado como instrumento para representar as referências - e mais genericamente qualquer tipo de relação - entre os documentos. O seu emprego facilita, em um ambiente de rede com recursos (documentos) distribuídos entre diversas fontes eletrônicas de informação, a construção de um hipertexto global entre os documentos jurídicos, legislativos e de bases de conhecimento contendo as relações entre os mesmos.

1.2. Associação do Nome Uniforme ao Documento

A associação do nome uniforme ao documento dá-se por meio de uma meta-informação, que pode ser:

- inserida no interior do próprio documento: é a solução natural a ser aplicada com a adoção do formato XML (*eXtensible Markup Language*), mas é factível também em outros formatos tais como HTML (*HyperText Markup Language*), PDF (*Portable Document Format*), DOC (Documento do Microsoft Word) ou ODF (*Open Document Format*);
- externa ao documento, mas estreitamente relacionada a este, de diversos modos: mesmo nome do arquivo e extensão predefinida, link hipertextual particular, atributo específico no banco de dados etc.

Tais modalidades são fortemente recomendadas no que se refere à implementação e à atualização automática dos catálogos (distribuídos ou centralizados), necessários ao processo de resolução, que contem, para todo documento, o nome uniforme e sua localização física.

Em perspectiva, deve-se tender à atualização automática, evitando-se a atualização manual dos catálogos, por causa do esforço necessário à sua manutenção e da intrínseca perda de validade com o tempo.

Em tais catálogos podem estar presentes também outras informações, como: metadados descritivos (ementa, título, assunto, dados da publicação oficial etc), relacionamentos com outros documentos e período de vigência.

1.3. Utilização do Nome Uniforme nas Referências

O nome uniforme será utilizado em larga escala nas referências de hipertexto realizadas a partir do texto integral de documentos ou a partir de sistemas de informações.

Tal vínculo pode ser realizado das seguintes maneiras:

- inserindo-se manualmente, no texto do documento que realiza a referência, o link com o nome uniforme: operação onerosa, sobretudo para os documentos já publicados em rede;
- construindo-se automaticamente (de maneira permanente ou provisória) o link com o nome uniforme, por meio de analisadores (*parser*) das referências no interior do texto: operação mais econômica mas sujeita a um certo percentual de imprecisão e incompletude. As referências não são de fato sempre formalmente corretas e completas, e devem por isso ser complementadas pelo operador ou interpretadas pelo usuário.
- construindo-se automaticamente (de maneira permanente ou provisória) o link com o nome uniforme, a partir de metadados (dados descritivos) existentes nos sistemas de informações dos diversos órgãos.

1.4. Tipos de URN

Considerando os cenários apresentados de uso da URN, podemos definir dois tipos de URN:

- URN Canônica – É a URN normalizada segundo o vocabulário e regras mais restritas, existindo apenas uma para cada entidade identificável (Documento, Documento Complexo e Fragmento). Ela é utilizada obrigatoriamente pelo provedor de informações (ver LexML Parte 4 – Coleta de Metadados), relacionada a um documento específico, devendo ser codificada sempre na forma completa.
- URN de Referência – É a URN utilizada pelos usuários ou sistemas de informação nas referências a documentos, podendo utilizar a forma compacta, omitir elementos que assumirão valores *default* ou ainda referenciar nomes alternativos.

A URN de Referência facilita a referência a documentos pois permite múltiplas formas de endereçamento. Por exemplo, para referenciar uma lei não é necessário informar o mês e o dia da data de assinatura (data representativa), bastando informar o ano; ao referenciar a norma “Lei Maria da Penha” não é necessário informar números e datas (ex: urn:lex:br:federal:lei:lei.maria.penha).

1.5. Princípios-base do Nome Uniforme (URN)

O nome uniforme deve ser unívoco, isto é, deve identificar uma e apenas uma entidade, e é construído de modo a ser, tanto quanto possível, alinhado com os princípios enumerados a seguir.

1.5.1. Princípios Gerais da URN

- auto-explicativo para os usuários;
- dedutível por meio de regras simples e claras;

- o alinhado constantemente aos demais padrões do projeto.

1.5.2. Princípios da URN Canônica

- o unívoco para o espaço das URNs canônicas;
- o registrável com o mínimo de informação (o suficiente para ser unívoca)
- o representativo dos aspectos formais do documento;
- o em conformidade, à data de sua emissão, com a estrutura/organização da autoridade emitente e com a tipologia do documento;

1.5.3. Princípios da URN de Referência

- o representativo dos aspectos formais e/ou substanciais do documento;
- o mapeável em URNs canônicas cabíveis;
- o compatível com a prática em uso para criar referências;
- o reduzido ao essencial, para simplificar os links com outros documentos;
- o gerado automaticamente por analisadores das referências no texto;

1.6. Características Gerais do Sistema de Identificação e Referência a Documentos

O sistema de referenciamento baseado em nomes uniformes deve prever:

- o um esquema de atribuição dos nomes capaz de representar de maneira unívoca qualquer documento normativo, emitido por qualquer autoridade a qualquer tempo (passado, presente e futuro);
- o um mecanismo para a resolução de um nome uniforme para a localização dos recursos correspondentes na rede.

O presente documento trata em detalhes apenas o primeiro dos requisitos indicados, isto é, a definição do esquema de atribuição dos nomes uniformes. A “Parte 5 – Serviço de Resolução” trata das diretrizes utilizadas pelo serviço de resolução.

2. Sintaxe Utilizada neste Documento

No presente documento foi utilizada a sintaxe típica de muitos RFCs da Internet, baseada na meta-linguagem BNF (*Backus-Naur Form*), e, em particular, naquela do documento RFC 2141 "URN Syntax".

As suas principais características são:

- o os elementos são contidos entre os sinais de menor e maior (" $<$ " e " $>$ ");
- o um elemento é separado da sua especificação com a *string* ":: $=$ ";
- o os elementos em alternativa são separados por uma barra vertical (" $|$ ");
- o as *strings* de caracteres são contidas entre aspas (" $''$ ");

- as partes opcionais são contidas entre colchetes ("[" e "]");
- um grupo de elementos é contido entre parênteses ("(" e ")");
- uma expressão que precede um elemento, ou um grupo de elementos, indica um fator de repetição, com os seguintes formatos:
 - `<n>` : `<n>` vezes;
 - `<n>*` : `<n>` ou mais vezes;
 - `*` : zero ou mais vezes.

Em cada diagrama de sintaxe, será indicado se as regras são específicas da URN Canônica ou de Referência ou se elas se aplicam a ambos os casos, indicado, nesse caso, com a palavra “Comum”.

3. Espaço de nomes (*namespace*)

O documento RFC 2141 "URN Syntax" (RFC 2141) define a sintaxe geral de um nome uniforme, que deve ter a forma:

Comum	<code><URN> ::= "urn:" <NID> ":" <NSS></code>
-------	---

onde `<NID>` é o identificador do Espaço de Nomes (*namespace*) e `<NSS>` é um nome-específico qualquer pertencente ao Espaço precedente.

O Espaço de Nomes identifica o contexto no qual estes são válidos e significativos. Para os documentos jurídicos define-se o Espaço de Nomes Uniformes com o identificador "lex":

Comum	<code><NID> ::= "lex"</code>
-------	------------------------------------

e assim as URNs relativas têm a forma:

Comum	<code><URN> ::= "urn:lex:" <NSS-lex></code>
-------	---

3.1. Internacionalização do espaço de nomes "lex"

A fim de permitir a adoção do Espaço de nomes "lex" em nível internacional, respeitando-se a autonomia e as especificidades de cada país, foi definido o elemento-base `<local>`, colocado dessa forma na URN:

Comum	<code><NSS-lex> ::= <local> ":" <NSS-lex-local></code>
-------	--

Esse elemento, que deve ser composto pelo menos pelo código IETF do país ou da super-região ("eu", no caso da União Européia), podendo ainda, a critério de cada país, ser subdividido de forma hierárquica de modo a contemplar o mapeamento de regiões, estados, municípios ou de outras unidades político-administrativas inerentes à sua organização. Em resumo:

Comum	<code><local> ::= <codigo-pais> [<detalhamento-local>]</code>
-------	---

	<code><codigo-pais> ::= 2<minuscula></code>
--	---

O código IETF para o Brasil é a *string* "br", e assim as URNs de documentos brasileiros têm a forma:

Comum	<code><URN> ::= "urn:lex:br:" <NSS-lex-local></code>
-------	--

Para o Brasil, definiu-se ainda que o elemento `<local>` possui uma estrutura hierárquica que reflete a jurisdição da autoridade emitente, desdobrando-se, quando necessário, em unidades da federação e municípios ou em regiões de órgãos do Poder Judiciário. A forma de composição desse elemento será vista adiante.

Apesar de a forma do elemento `<NSS-lex-local>` ser de livre definição pelo país usuário do espaço de nomes, optamos por utilizar para o `<NSS-lex-br>` um esquema similar ao modelo do projeto italiano NIR (*Norme in Rete*), aproveitando-nos da experiência e de conceitos já desenvolvidos, testados e estabelecidos.

4. Sintaxe Geral do Nome-Específico Brasileiro

4.1. Caracteres Permitidos e Proibidos

São definidos de acordo com a RFC 2141 "*URN Syntax*", em especial figuram como não admitidos, além de uma série de caracteres especiais, todos aqueles nacionais (código ASCII superior ao 127), que devem ser convertidos na sequência de escape ("% <hex> <hex>") correspondente.

Por razões de várias naturezas, ilustradas a seguir, no nome-específico (`NSS-lex`) do namespace "lex" é permitido apenas este subconjunto de caracteres:

Comum	<pre> <NSS-lex-local> ::= 1*<aceitos-lex> <aceitos-lex> ::= <normal> <reservado> <detalhaID> <outro> <normal> ::= <minuscula> <numeral> "." <normalID> ::= <normal> <detalhaID> <minuscula> ::= "a" "b" "c" "d" "e" "f" "g" "h" "i" "j" "k" "l" "m" "n" "o" "p" "q" "r" "s" "t" "u" "v" "w" "x" "y" "z" <numeral> ::= "0" "1" "2" "3" "4" "5" "6" "7" "8" "9" <reservado> ::= ":" "@" ";" "," "[" "]" "!" "~" <detalhaID> ::= "_" "-" <outro> ::= "=" "\$" </pre>
-------	---

Os outros caracteres são eliminados ou convertidos, de acordo com as regras que serão apresentadas nas próximas seções.

A sintaxe completa dos nomes uniformes do Espaço "lex-br" é descrita nos Anexos A e B, onde são apresentadas as gramáticas da URN Canônica e da URN de Referência respectivamente.

4.2. Caracteres Reservados

São os caracteres que têm um significado especial na formação das URI (*Uniform Resource Identifier*) ou na formação de nomes específicos no Espaço "lex", e, portanto, devem ser empregados sempre e apenas para o escopo definido. Pertencem à primeira categoria os caracteres "%", "/", "?", "#" que conservam o significado típico das URLs, com exceção do caractere "#", que, no âmbito do LexML, não deve ser utilizado para indicar um "fragmento" do documento no nome uniforme, pois, para isso, existe um caractere específico ("\$").

Por sua vez, são reservados no nome-específico do Espaço "lex" os caracteres apresentados na Tabela 1.

Tabela 1. Caracteres Reservados.

Caractere(s)	Aplicação
" : "	separador dos elementos principais do nome.
" ; "	marcador da introdução de uma especificidade ou de um nível hierarquicamente inferior do elemento.
" @ "	separador de versão.
" ~ "	separador para indicar a forma da expressão do conteúdo (ex: texto, imagem etc.) e língua.
" ! "	separador de fragmento.
" [" , "] "	delimitadores utilizados para especificar intervalo de valores. Os valores de limite inferior e superior são separados pelo caractere " , " .
" , "	É utilizado nas seguintes situações: <ul style="list-style-type: none">▪ separador de múltiplas ocorrências de valores para um mesmo elemento principal, ao mesmo nível de especificidade (ex.: numeração e autoridade múltipla);▪ separador dos valores limites de um intervalo; e▪ separador do identificador de componente do título do componente (ex: anexo.1, tratado.cooperacao.amazonica).

4.3. Casos Especiais

4.3.1. Caracteres Maiúsculos e Minúsculos

O nome-específico <NSS> da URN, como as URLs, é sensível à grafia maiúscula ou minúscula.

A resolução dos nomes pertencentes ao Espaço "lex" deve ser independente da grafia, mas, para desfrutar plenamente das possibilidades de *caching* (memorização temporária para acesso rápido) de recursos, o nome-específico é construído sempre em caracteres minúsculos.

(ex.: "Ministério" é representado como "ministerio")

4.3.2. Caracteres Nacionais e Sinais Diacríticos

Mesmo sendo possível representar os caracteres nacionais e os sinais diacríticos por meio de uma seqüência de escape ("%<hex> <hex>"), por razões de simplicidade, de gestão e de comunicação estes caracteres especiais são excluídos do Espaço "lex" e, portanto, devem ser convertidos aos caracteres base.

(ex.: "saúde" é representada como "saude")

4.3.3. Substituição de Espaços, Conectivos e Pontuação

Na construção do nome-específico são eliminados da sentença correspondente todos os conectivos lingüísticos (artigos, preposições simples e com artigos, conjunções etc.), a pontuação e os caracteres especiais como apóstrofes, hífens, aspas etc.

As palavras significativas restantes são ligadas entre si por um ponto (".") em substituição ao espaço.

(ex.: "Ministério do Planejamento, Orçamento e Gestão" é representado como "ministerio.planejamento.orcamento.gestao")

4.3.4. Expansão das Abreviações

No nome-específico os nominativos são representados por extenso, e, portanto, devem-se expandir as abreviações de quaisquer tipos presentes na indicação das instituições (ex.: Min.), das estruturas (ex.: Gov.) ou das normas (ex.: circ.).

(ex.: "Min." é representado como "ministerio")

4.3.5. Expansão das Siglas

Para evitar possíveis homônimos provocados pelo uso de siglas na indicação das instituições, das estruturas ou das normas, no nome-específico essas devem ser explicitadas com as correspondentes formas extensas.

(ex.: "SRF" é representada por "secretaria.receita.federal")

4.3.6. Elementos e Números Ordinais

Na construção do nome-específico os eventuais números ordinais presentes nos elementos são sempre representados por algarismos arábicos, independente da grafia

que assumam: número romano, alíneas, adjetivo, letras, número arábico com ápice etc. (IV, alínea b, terceiro, 1º, 2ª etc.).

Identificadores de fragmentos de documento (ex.: artigos, incisos, capítulos etc.) possuem sintaxe compacta própria, conforme será apresentado nas seções seguintes.

(ex.: "Anexo III" é representado como "anexo.3")

"Anexo D" é representado como "anexo.4"

"Art. 11" é representado como "art11"

"Art. 12-B" é representado como "art12-2"

"Inciso III do Art. 12-B" é representado como "art12-2_inc3")

4.4. Formato da Data

Para tornar o nome uniforme mais facilmente ordenável cronologicamente e mais legível aos operadores, as datas são representadas sempre em forma numérica e no formato:

Comum	<pre><data> ::= <ano> "-" <mes> "-" <dia> <ano> ::= 4<numeral> <mes> ::= 2<numeral> <dia> ::= 2<numeral></pre>
-------	--

(ex.: "2 de setembro de 99" é representado por "1999-09-02")

4.4.1. Tratamento de Datas Múltiplas e Período

Alguns atos, mesmo sendo únicos, são identificados por mais de uma data; nesse caso, no campo <datas> devem ser relacionadas todas as datas presentes, separadas por “,”.

Caso seja necessário especificar um intervalo fechado entre duas datas, utiliza-se os delimitadores “[” e “]” com as datas separadas por “,”.

Dessa forma, o elemento <datas> possui a seguinte definição:

Comum	<pre><datas> ::= (<data> *["," <data>]) <periodo> <periodo> ::= "[" <data> "," <data> "]"</pre>
-------	--

(ex.: "1999-03-02,1999-04-15,2000-08-01" e
"[1999-03-02,2000-08-01]")

5. Construção do Nome-Específico (NSS-lex-br)

5.1. Estrutura do Nome-específico <NSS-lex-br>

Considerando o Modelo de Referência do Projeto LexML Brasil (Parte 1), são três os tipos de entidades que podem ser endereçadas por uma URN:

- Documento Complexo (*F15 Work Complex*) – representa o mais alto nível de abstração do modelo e referencia o conteúdo do documento (matéria) no tempo, não se preocupando em especificar detalhes de versão ou forma de expressão. Normalmente é esta a entidade referenciada pelo usuário quando da pesquisa de uma norma.
- Documento Individual (*F14 Individual Work*) – representa o conteúdo abstrato (matéria) da versão específica de um documento. Quando um Documento Individual é base para derivação de um outro Documento Individual (como no caso da publicação), ambas as instâncias passam a ser membros de um Documento Complexo. Cada instância de Documento Individual possui uma expressão auto-contida relacionada (*F22 Self Contained Expression*) que pode conter, de forma recursiva, outras expressões, denominadas fragmentos.
- Fragmento de Documento Individual (*F22 Self Contained Expression*) – representa a partição de um documento individual. No caso de normas jurídicas ou proposições legislativas, pode ser um artigo, um dispositivo que agrupa artigos (ex. título, capítulo, seção etc.) ou ainda um dispositivo de detalhamento de um artigo (ex. alínea, inciso etc.). No caso de acórdãos, pode ser um dos seus elementos constituintes (ex. relatório, voto etc.)

Em nosso ordenamento jurídico, o conteúdo normativo expressa-se como um texto na língua portuguesa. No entanto, em alguns casos, pode haver variação tanto na língua utilizada quanto na forma de expressão (ex: mapas, figuras etc).

Portanto, a estrutura do nome-específico de cada entidade é representada por:

Comum	<pre> <NSS-lex-local> ::= <documentoComplexo> (<documentoIndividual> ["!" <fragmento>]) <documentoIndividual> ::= <documentoComplexo> ["@" <versao>] ["~" <forma>] <documentoComplexo> ::= <documento> </pre>
-------	---

5.2. Nomes Alternativos (alias)

A um mesmo documento podem ser atribuídos diversos nomes uniformes, a fim de favorecer a ligação – por referenciamento e por pesquisa - com outros documentos. Essa facilidade pode ser utilizada para aqueles documentos que, devido à sua individualidade, acabam sendo referenciados na prática sob ângulos diversos, como, por exemplo, a forma de promulgação e o conteúdo específico (ex.: Regimento Interno veiculado por uma Resolução).

É muito comum também a referência de normas utilizando apelidos, tais como ‘Código Civil’, ‘Código de Defesa do Consumidor’, ‘Lei do Audiovisual’ etc. A codificação de nomes alternativos será realizada utilizando valores especiais para os elementos <tipo-documento> e <descriptor>, que serão apresentados nas próximas seções.

6. Elemento <local>

O elemento <local> delimita a localidade na qual será definida a autoridade emitente do documento. É importante ressaltar que esse elemento não está relacionado ao conteúdo e sim à autoridade emitente. Por exemplo, para a autoridade “camara.municipal” o elemento <local> indicará o nome do município no qual a câmara municipal está localizada.

O elemento <detalhamento-local> tem a seguinte estrutura:

URN Canônica	<pre> <local> ::= <codigo-pais> [<detalhamento-local>] <detalhamento-local> ::= ([";" <unidade-federacao> [";" <municipio>]]) ([";" <unidade-federacao>] [";" <local-judiciario>]) <detalhe-ramo> ::= (("regiao." "zona.") 1* <numeral>) ("secao.judiciaria" "comarca") (<unidade-federacao> *[", " <unidade-federacao>]) (<municipio> *[", " <municipio>])) </pre>
URN de Referência	<pre> <detalhamento-local> ::= ([";" <unidade-federacao-ou-uf> [";" <municipio>]]) ([";" <unidade-federacao-ou-uf>] [";" <local-judiciario>]) <detalhe-ramo> ::= (("regiao." "zona.") 1* <numeral>) ("secao.judiciaria" "comarca") (<unidade-federacao-ou-uf> *[", " <unidade-federacao-ou-uf>]) (<municipio> *[", " <municipio>])) <unidade-federacao-ou-uf> ::= <unidade-federacao> 2<minúscula> </pre>
Comum	<pre> <unidade-federacao> ::= 1*(<minuscula> ".") <municipio> ::= 1*(<minuscula> ".") <local-judiciario> ::= <ramo-justica> *[";" <detalhe-ramo>] <ramo-justica> ::= "justica." ("federal" "trabalho" "eleitoral" "militar" "estadual") </pre>

Os elementos <unidade-federacao> e <municipio> representam o nome da unidade da federação e do município, respectivamente, ambos grafados por

extenso em minúsculas, sem abreviaturas, conectivos, apóstrofo, sinais de pontuação ou espaços. Nas URNs de Referência, poderá ser informada a sigla da unidade da federação, ao invés do nome por extenso. Na URN Canônica, o nome da unidade da federação deve ser grafado por extenso.

A segunda forma de composição do elemento <detalhamento-local> especifica a jurisdição dos órgãos do Poder Judiciário, considerando os vários ramos de especialização da justiça brasileira. É importante ressaltar que o nome da unidade da federação precede o nome do ramo da justiça apenas no caso da “justiça estadual”.

Exemplos de valores para o elemento <local>:

Jurisdição:	Representação (elemento <local>):

Brasil	br
Pindamonhangaba-SP	br;sao.paulo;pindamonhangaba
João Pessoa-PB	br;paraiba;joao.pessoa
Estado de São Paulo	br;sao.paulo
Distrito Federal	br;distrito.federal
São Paulo-SP	br;sao.paulo;sao.paulo
São João do Meriti-RJ	br;rio.janeiro;sao.joao.meriti
TSE	br;justica.eleitoral
TRE da Paraíba	br;justica.eleitoral;paraiba
TRT da 13ª Região	br;justica.trabalho;regiao.13
Zona 22 do TRE-PB	br;justica.eleitoral;paraiba;zona.22
Comarca de Macapá-AP	br;amapa;justica.estadual;comarca;macapa
Seção Judiciária Goiás	br;justica.federal;regiao.1;secao.judiciaria;goias

7. Elemento <documento>

O elemento <documento> deve conter informações adequadas a identificar univocamente as entidades referentes aos documentos jurídicos e legislativos, as quais são essencialmente: a autoridade emissora, o tipo de provimento e o descritor (datas, números, etc.).

A estrutura do identificador do documento é composta por três elementos fundamentais, claramente distintos entre si e de acordo com uma ordem que identifica domínios e competências sempre mais restritas:

Comum	<documento> ::= <autoridade> ":" <tipo-documento> ":" <descriptor>
-------	--

Os elementos principais do nome uniforme são geralmente articulados em diversos componentes elementares, para os quais são fixadas regras de representação (critérios, modalidade, sintaxe e ordem).

8. Elemento <autoridade>

8.1. Indicação da Autoridade

O elemento <autoridade> indica, de acordo com o caso:

- a efetiva autoridade emissora da norma, isto é, aquela que realmente adotou ou da qual emanou o ato;
- a instituição por meio da qual o ato é registrado, conhecido e referenciado, ainda que emitido ou de alguma forma produzido por outros sujeitos: é o caso, por exemplo, dos projetos de leis que são identificadas com a Casa Legislativa na qual são apresentadas e não com seus autores e subscritores (parlamentares, poder executivo etc.);
- a instituição regulamentada (e indicada nas referências) pela norma, ainda que emitida por outra autoridade: é o caso, por exemplo, do estatuto de um ente. Nesta hipótese o nome uniforme será um alias do ato de adoção ou emanção.

8.2. Pluralidade dos Sujeitos

Alguns atos têm diversas autoridades emittentes, quando emanadas de uma pluralidade de sujeitos (ex.: portaria interministerial, acordos etc.). Neste caso, no elemento <autoridade> são representados, adequadamente separados, todos os sujeitos envolvidos segundo a forma:

Comum	<autoridade> ::= <sujeito> *["," <sujeito>]
-------	--

(ex.: "ministerio.fazenda,ministerio.justica").

Cabe ressaltar que, no caso de diversas autoridades emittentes, de forma independente da seqüência em que os nomes aparecem no documento, deve-se utilizar a ordem alfabética dos nomes codificados dos sujeitos na composição do nome uniforme. Em uma URN de referência, podem-se especificar os nomes dos sujeitos em qualquer ordem, pois, no processo de resolução, eles serão convertidos para a ordem alfabética.

(ex.: "ministerio.justica,ministerio.fazenda"

e "ministerio.fazenda,ministerio.justica" referenciam, de forma equivalente, a mesma autoridade emittente).

8.3. Modalidade de Indicação do Sujeito

Cada autoridade emissora (aquela que produz o ato) ou referenciada (aquela que é regulamentada pelo ato) consiste essencialmente em um cargo institucional (ex.: Presidente da República) ou em uma instituição (ex.: Presidência da República). Quando for necessária, para uma identificação precisa da norma, a indicação de um órgão no interior da instituição, esta é efetuada segundo a ordem hierárquica do geral para o particular, partindo da instituição, seguindo os vários níveis da estrutura

(Secretaria, Departamento etc.) e terminando com o cargo relativo (Presidente, Diretor, Secretário etc.). Portanto, a estrutura do sujeito emite se fica sendo:

Comum	<pre> <sujeito> ::= (<instituicao> *[";" <orgao>] [";" <funcao>]) <cargo> <instituicao> ::= 1*<normal> <orgao> ::= 1*<normal> <funcao> ::= 1*<normal> <cargo> ::= 1*<normal> </pre>
-------	--

(ex.: "ministerio.fazenda;secretaria.receita.federal;secretario")

8.3.1. Indicação do órgão

Em alguns tipos de norma, mesmo a função da autoridade do órgão emite se sendo univocamente determinada (ex.: Secretário da SRF), geralmente esta não é indicada nas referências. Nesses casos, em analogia ao que é adotado na prática, a indicação da autoridade emite se é limitada ao mínimo indispensável em função do tipo do ato.

(ex.: "ministerio.fazenda;secretaria.receita.federal:instrucao.normativa")

e não "ministerio.fazenda;secretaria.receita.federal;secretario:instrucao.normativa")

8.3.2. Indicação da função

Em linhas gerais o componente <funcao>:

- o é indicado, às vezes no lugar do próprio órgão, quando se trata de um cargo do tipo "político", representativo ou eletivo, ou caso seja natural ao tipo de ato

(ex.: "universidade.brasilia;reitor:oficio")

e não "universidade.brasilia;reitoria:oficio");

- o é indicado quando se trata de uma figura maior da instituição (ex.: diretor geral, secretário geral, diretor administrativo etc.), à qual nem sempre é possível associar uma estrutura interna específica

(ex.: "ministerio.justica;departamento.policia.federal;diretor.geral");

- o não é indicado quando corresponde à responsabilidade de uma estrutura (tipicamente preenchida por um funcionário ou por um dirigente); nesse caso indica-se apenas a estrutura e não o dirigente

(ex.: "ministerio.justica;departamento.administracao.penitenciaria")

e não "ministerio.justica;departamento.administracao.penitenciaria;chefe").

Por outro lado, a função deverá ser indicada quando:

- o não coincide com a direção ou a responsabilidade da estrutura (ex.: subsecretário, vice-diretor etc.);
- o o tipo de ato possa ser de origem tanto monocrática quanto colegiada: em tal caso a indicação do cargo serve para eliminar a ambigüidade.

8.3.3. Convenções para a autoridade de documentos normativos

Por convenção, as normas do tipo constituição, lei orgânica, emendas, leis complementares, leis delegadas, medidas provisórias, decretos-leis e decretos têm a autoridade indicada apenas com os termos "federal", "estadual" ou "municipal", dependendo da esfera pública.

(ex.: "urn:lex:br:**federal**:emenda.constitucional"

e não "urn:lex:br:congresso.nacional:emenda.constitucional",

"urn:lex:br;sao.paulo;campinas:**municipal**:decreto"

e não "urn:lex:br;sao.paulo;campinas:prefeitura:decreto")

"urn:lex:br;sao.paulo;campinas:**municipal**:lei"

e não "urn:lex:br;sao.paulo;campinas:camara.municipal:lei")

As resoluções, portarias, instruções normativas e outras normas de hierarquia inferior devem especificar a autoridade emitente de forma unívoca.

9. Elemento <tipo-documento>

9.1. Critérios para a Indicação do Tipo do Documento

A indicação obrigatória da autoridade emitente nos nomes uniformes torna supérflua, na grande maioria dos casos, a qualificação adicional da norma (lei municipal, portaria ministerial etc.), mesmo se não sistematicamente, pelo menos na prática.

Quando a combinação autoridade-norma identifica claramente o documento em questão, o tipo de ato não é qualificado por meio de atributos que se refiram à autoridade emitente.

(ex.: "urn:lex:br;sao.paulo;ribeirao.preto:municipal:lei"

e não "urn:lex:br;sao.paulo;ribeirao.preto:municipal:lei.municipal")

9.2. Especificação Adicional do Tipo de Documento

Para a identificação do ato no elemento <tipo-documento>, geralmente é suficiente a indicação da sua tipologia. Todavia, é bastante difundida na prática a utilização de siglas que identificam um subtipo do ato (ex.: PEC – Proposta de Emenda Constitucional; ADI – Ação Direta de Inconstitucionalidade). Nesses casos, o tipo de documento pode ser seguido por especificações, de modo a viabilizar a identificação, conforme apresentado a seguir:

Comum	<pre> <tipo-documento> ::= ((<tipo-norma> <tipo-jurisprudencia> <tipo-projeto-norma>) [";" <nome-subtipo-sequenciador >]) ("publicacao.oficial;" <nome-periodico-oficial> [";" <nome-secao-periodico-oficial>] [";" <detalhe-extra-suplemento>]) <tipo-projeto-norma> ::= 1*<normal> <tipo-norma> ::= 1*<normal> <tipo-jurisprudencia> ::= 1*<normal> <nome-subtipo-sequenciador> ::= 1*<normal> <nome-periodico-oficial> ::= 1*<normal> <nome-secao-periodico-oficial> ::= 1*<normal> <detalhe-extra-suplemento> ::= ("edicao.extra" "suplemento") ["." 1*<numeral>] </pre>
-------	---

Os possíveis valores para `<tipo-norma>`, `<tipo-jurisprudencia>` e `<tipo-projeto-norma>` são definidos em vocabulários específicos.

O detalhamento `<nome-subtipo-sequenciador>` identifica o subtipo do documento que foi utilizado na seqüência de numeração dos documentos no contexto dos sistemas de informação da autoridade emitente. Os valores são especificados no vocabulário controlado utilizando o contexto da localidade e da autoridade emitente. Por exemplo, o valor “plc”, no contexto do Senado Federal, é definido como um subtipo de “projeto.lei.complementar”. Nesse caso específico, pode-se utilizar a sigla diretamente na codificação da URN Canônica.

9.3. Especificação da Publicação Oficial

O tipo "publicacao.oficial" permite identificar um periódico oficial que publica os textos das normas, julgados e proposições legislativas. Ao codificar informações sobre a publicação oficial, deve-se ser o mais específico possível, devendo-se codificar o nome da publicação oficial por extenso e, se existente, a identificação da seção. No caso de edições-extras e suplementos, deve-se utilizar o elemento `<detalhe-extra-suplemento>`.

(ex: “publicacao.oficial;diario.oficial.uniao;secao.1” ou

“publicacao.oficial;diario.justica.eletronico” ou

“publicacao.oficial;diario.oficial.uniao;secao.1;suplemento”)

10. Elemento `<descriptor>`

10.1. Indicação do Descritor

A codificação do descritor do documento geralmente constitui-se de uma data representativa (assinatura no caso de normas, iniciativa para proposições legislativas e data de julgamento para os julgados) e um número de identificação (seqüencial, número de protocolo etc.). No caso de componentes adicionais do documento, como, por exemplo, anexos, adicionam-se os descritores destes componentes separados por “;”.

No que se refere à data representativa, deve-se observar que algumas normas apresentam datas múltiplas. Em relação ao número do documento, este pode não existir de fato (ex.: atos não numerados) ou ainda ser múltiplo (ex.: atos unificados). Por essa razão, prevê-se que os elementos de codificação possam ter valores múltiplos. Nestes casos, utiliza-se o caractere “,” para delimitar cada valor.

Para efeito de implantação da Rede de Informações LexML, considerando que muitos sistemas de informações não possuem o registro da data na sua forma completa, será permitido informar apenas o ano da data representativa na URN Canônica. No futuro, essa permissão poderá ser revista, tornando, por exemplo, obrigatória a especificação da data completa para os documentos cuja data representativa seja maior do que uma data específica. Recomendamos o uso da data completa, sendo esta obrigatória nos casos dos documentos não numerados. No caso das URNs de Referência, sempre será possível abstrair o dia e o mês da data representativa.

De acordo com a estrutura global do nome uniforme, do geral ao particular, define-se como formato de indicação a seqüência “data número”, na forma:

URN Canônica	<pre> <descriptor> ::= <datasOuAno> ";" <identificadores> *[";" <id-componente> ["," <titulo-componente>]] [";"retificacao." <seq-retificacao>] </pre>
URN de Referência	<pre> <descriptor> ::= ((<datas> [";" <identificadores>]) (<datasOuAno> ";" <identificadores>) ([<datasOuAno> ";"] <apelido-documento>)) *[";" <id-componente> ["," <titulo-componente>]] [";"retificacao." <seq-retificacao>] </pre>
Comum	<pre> <id-componente> ::= 1*<normal> <titulo-componente> ::= 1*<normal> <seq-retificacao> ::= 1*<numeral> <datasOuAno> ::= <datas> <ano> </pre>

(ex.: "2000-12-06;126",

"2003-10-01;10741",

```
"2003-10-01;10741;anexo.1",
"2003-10-01;estatuto.idoso",
"1993-07-20;8685" e
"1993-07-20;lei.audiovisual" )
```

10.2. Elemento <identificadores>

O número do documento é representado pelo elemento <id-documento>. Alguns documentos podem ser identificados por uma série de números. Nesse caso, relacionam-se no campo <identificadores> todos os números.

O elemento <identificadores> possui a seguinte estrutura:

Comum	<pre><identificadores> ::= (<id-documento> *["," <id-documento>]) (<numero-lex>) (<numero-seq>) <numero-lex> ::= "lex-" 1*<numeral> <numero-seq> ::= "seq-" <sigla-orgao> "-" 1*<numeral> <sigla-orgao> ::= 1*<normal> <id-documento> ::= 1*<normalID></pre>
-------	---

(ex. "2000-06-12;c-10-97,c-11-97,c-12-97")

Os caracteres não permitidos (ex.: "/") ou reservados (ex.: ":") , incluindo a vírgula, não podem estar presentes no interior do <id-documento>, e portanto são convertidos em "-".

Os elementos <numero-lex> e <numero-seq> são utilizados para identificação de documentos sem numeração, conforme especificação apresentada na próxima seção.

10.2.1. Documentos sem Numeração

Os documentos sem numeração têm uma referência não unívoca, já que podem existir diversos documentos do mesmo tipo, emitidos no mesmo dia pela mesma autoridade.

Dado que um nome uniforme deve identificar univocamente cada um deles, o campo <descriptor> deve conter um elemento discriminante. A unicidade deste número seqüencial será gerada pelo sistema LexML considerando a localidade, a autoridade emitente, o tipo de documento e a data de assinatura, além dos metadados descritivos disponíveis, considerando, sempre que possível, a ordem em que o documento aparece na publicação oficial. Para tornar este identificador gerado claramente distinguível, tal número é precedido pela *string* "lex-".

Comum	<pre><numero-lex> ::= "lex-" 1*<numeral></pre>
-------	--

(ex. "br:federal:decreto:1999-12-21;lex-16")

Caso o provedor de dados utilize um seqüencial de identificação diverso daquele utilizado pelo LexML, poderá indicar o número do descritor próprio, qualificado pela sigla do órgão (conforme vocabulário controlado), utilizando a seguinte codificação:

Comum	<pre><numero-seq> ::= "seq-" <sigla-orgao> "-" 1*<numeral> <sigla-orgao> ::= 1*<normal></pre>
-------	---

(ex. "urn:lex:br:federal:decreto:1999-12-21;seq-sf-32")

Rotinas de conciliação poderão relacionar os descritores `<numero-lex>` e `<numero-seq>` correspondentes, tendo por base os metadados descritivos.

Caso ocorra, por erro, a atribuição de múltiplos identificadores `<numero-lex>` para um mesmo documento, deve-se utilizar o recurso de equivalência entre identificadores para aglutinar os múltiplos registros de identificação do documento.

10.2.2. Indicação do Número de Identificação do Documento

Para facilitar a construção automática do nome uniforme a partir das referências, o número de identificação do ato é representado exatamente na forma que foi atribuído pela autoridade emitente, considerando as conversões necessárias de caracteres especiais.

(ex.: um ofício com a numeração AIPA/CR/35 de 06/11/2001 teria como elemento `<descriptor>` do nome uniforme: "2001-11-06;aipa-cr-35").

De acordo com as regras de numeração adotadas pelo emanante, este pode aceitar a presença de informações redundantes (ex.: autoridade, ano de emissão etc.) já presentes em outras partes do nome uniforme (na mesma forma ou em forma diversa).

Por outro lado, a eliminação automática de tais redundâncias a partir das citações que trazem o número de maneira completa (ex.: ofício AIPA/CR/35) não é uma operação confiável, já que existe uma variabilidade enorme nas regras de numeração adotadas pelas autoridades emitentes.

Tendo em conta que as referências podem ser efetuadas também de forma reduzida (ex.: ofício AIPA no. 35 de 2001), o que faz com que o nome uniforme construído também seja reduzido, no final do processo de resolução o número identificador é considerado como sendo formado por diversos componentes elementares, cada um deles delimitado por caracteres especiais.

10.2.3. Esclarecimentos Sobre os Números de Identificação do Documento

Na sintaxe dos nomes uniformes, o número singular (`<id-documento>`) do elemento `<identificadores>` é, no seu todo, uma *string* opaca que, combinada com os outros elementos, forma um identificador unívoco.

Na numeração, são relacionadas na mesma ordem todas as informações presentes no documento que caracterizam a sua identificação, mesmo que algumas (ex.: ano) já estejam contidas em outros elementos (ex.: data).

Assim, por exemplo, se o identificador de um ofício é 409/98, utiliza-se a numeração completa "409-98" e não apenas 409, mesmo que o ano 1998 já seja um conteúdo da data.

Não existe regra pela qual os componentes devam ser representados no interior de uma numeração em uma determinada ordem (ex.: primeiro o número sequencial e depois o ano, ou vice-versa): assim, por exemplo, o nº 409/98 é reportado como "409-98", o nº 98/409 como "98-409". Os dois números "409-98" e "98-409" são ambos válidos e identificam, considerando a paridade dos outros elementos, dois documentos distintos.

Assim, utiliza-se exatamente a forma adotada pela autoridade emitente, segundo a prática difundida no seu interior, para uma classe (ou período) de atos, eventualmente distinta de outras classes ou períodos (ex.: para os ofícios: número/ano; para os memorandos: ano/número).

No caso de normas jurídicas que utilizem as autoridades emittentes convencionadas para cada esfera administrativa ("federal", "estadual" ou "municipal"), não se deve codificar o ano da norma no elemento `<id-documento>`.

É prevista ainda a presença de diversos números identificadores do documento; por princípio, tais números seriam da mesma natureza, atribuídos, assim, com base em um registro comum (ex.: reunificação de projetos de lei, de casos etc.) e, desse modo, congruentes entre si. Em tal hipótese, a ordem não tem relevância, mas é o conjunto dos números singulares que distingue o ato.

Se o documento tem diversos números, estes podem ser representados na mesma seqüência em que aparecem no documento: no esquema é irrelevante a ordem na qual os números singulares são representados (crescente, decrescente, nenhuma) para a unicidade do nome uniforme (ex.: "135,409" e "409,135" são ambos válidos, e identificam, considerando-se a paridade dos outros elementos, o mesmo documento).

10.3. Identificador de Componentes de um Documento

O elemento `<id-componente>` deve ser utilizado nos casos em que um documento é acompanhado de outros documentos (ex.: ocorrência de anexos). Este identificador é formado por um descritor curto acompanhado de um número sequencial (ex.: anexo.1) seguido, sempre que possível, de um título (`<titulo-componente>`) pelo qual o documento é conhecido.

(ex: decreto:1980-08-18;85080;anexo.1,tratado.cooperacao.amazonica
e não decreto:1980-08-18;85080;anexo.1)

10.3.1. Anexos Dependentes Funcionais e Formais

A presença de anexos de várias naturezas (anexos, tabelas, mapas etc.) é muito freqüente nos documentos legislativos e jurídicos e não é raro o caso no qual esses assumem a forma de textos articulados (ex.: regulamentos). Apesar de quase sempre serem partes integrantes do documento normativo principal, os anexos podem ser referenciados (e mesmo sofrer modificações) de maneira independente do ato a que pertencem. Torna-se, assim, necessário identificar univocamente não apenas o documento principal, mas também cada um dos seus anexos.

Os anexos dependentes formais são subdivisões efetivamente previstas no documento principal com o objetivo de tornar o seu texto mais fluente e organizado; esses são claramente nominados ou facilmente identificáveis no ato (ex.: “Anexo I”) e geralmente referenciados no mesmo texto (ex.: "no Anexo I encontra-se a lista ...").

Já os anexos dependentes funcionais são, ao contrário, partes integrantes do próprio documento principal, que têm, no entanto, uma forma de registro não-textual e que por isso, na representação eletrônica do documento, precisam ser forçosamente segregados de forma a constituir objetos separados. O documento completo será recomposto na fase de apresentação. Isso acontece regularmente na presença de objetos multimídia (ex.: mapas geográficos, imagens, fotografias etc.), mas podem apresentar-se também como textos registrados em formato eletrônico não textual (por ex.: a imagem digitalizada da versão em língua estrangeira de um acordo internacional). É necessário indicar, no elemento que trata da forma da expressão, a natureza do conteúdo, de acordo com o respectivo vocabulário controlado.

10.3.2. Identificação do Anexo

Ao anexo de qualquer natureza é sempre atribuído um nome uniforme distinto, independente do seu valor autônomo e da sua forma de registro (junto ou separado do documento principal). O seu nome uniforme é formado pelo nome uniforme do documento ao qual está vinculado e por uma desinência identificadora do próprio anexo.

Para os anexos dependentes formais, a desinência traz a denominação oficial do anexo e eventuais especificações adicionais (ex.: título), para facilitar a recuperação nos casos de falta do identificador.

(ex.: "urn:lex:br:ministerio.fazenda;secretaria.receita.federal:instrucao.normativa:2007-02-12;25;anexo.a,tabela.aliquota.imposto.renda.pessoa.fisica")

Para os anexos dependentes funcionais, na falta de uma denominação oficial, utilizam-se diretamente as informações que ajudam na sua identificação, como, por exemplo, a legenda.

(ex.: se no interior da instrução normativa anterior fosse apresentado um gráfico com a evolução da arrecadação, seria criado um anexo funcional com o nome uniforme "urn:lex:br:ministerio.fazenda;secretaria.receita.federal:instrucao.normativa:2007-02-12;25;grafico.evolucao.arrecadacao")

Os caracteres não admitidos (ex.: "/") ou reservados (ex.: ":") não podem estar presentes dentro do `<id-componente>`, e, portanto, deverão ser substituídos por ".".

10.3.3. Anexos de anexos.

Nos casos em que um anexo contenha um ou mais anexos, os identificadores correspondentes serão construídos adicionando-se ao identificador do anexo-pai os dos anexos vinculados.

(ex.: Supondo-se que o gráfico do exemplo anterior estivesse contido em um anexo específico da Instrução Normativa nº 25 que tratasse apenas da justificativa para a correção da tabela do IRPF, o nome uniforme para esse gráfico seria: "urn:lex:br:ministerio.fazenda;secretaria.receita.federal:instrucao.normativa:2007-02-12;25;anexo.b.justificativa.correcao;grafico.evlucao.arrecadacao").

10.3.4. Descritor de Retificação

Nos casos do texto da retificação de um documento, deve-se, após a qualificação do documento no elemento descritor, informar o termo “retificacao.” seguido de um seqüencial de ordem da retificação do documento. Por sua vez, o elemento `<versao>` deve apresentar três sub-elementos:

1. a data da versão do ato a que a retificação se refere, que, no caso de normas, é a data de início da vigência;
2. o nome do evento “publicacao”; e
3. a data da publicação da retificação.

(ex.: lei:1990-09-11;8078;retificacao.1@1990-09-12;publicacao;2007-01-10).

10.4. Descritor da Publicação Oficial

O elemento descritor, no caso do identificador da publicação oficial, contém o número da edição, conforme exemplo abaixo:

(ex.:urn:lex:br:imprensa.nacional:publicacao.oficial;diario.oficial.uniao;secao.1:2008-07-22;139)

No caso de não existir o número da edição, deve-se aplicar a regra especificada para documentos não numerados.

(ex.:urn:lex:br:congresso.nacional:publicacao.oficial;diario.congresso.nacional:2003-02-01;seq-cn-0)

11. Elemento `<fragmento>`

A possibilidade de identificar univocamente o fragmento de um documento principal ou de seu anexo é importante, pois permite estabelecer relacionamentos mais precisos, como nos casos dos relacionamentos entre dispositivos e normas. Por exemplo, a regulamentação do artigo de uma lei pode ser realizada por um decreto.

Nesse caso, ao invés de relacionar as normas deve-se estabelecer relacionamento entre o dispositivo da lei e o decreto no todo.

O elemento `<fragmento>` deve ser utilizado apenas na identificação das partições internas de uma expressão, como, por exemplo, no caso de uma norma jurídica, na identificação de artigos, dispositivos que agrupam artigos (títulos, capítulos etc.) ou detalhamento de artigos (parágrafo, inciso, alínea etc.). Os elementos de identificação das partições internas devem seguir as regras de nomenclatura previstas no XML Schema do Projeto LexML (Parte 3).

Comum	<pre> <fragmento> ::= (<id-particao> <intervalo-ids>) *[" , " <fragmento>] <intervalo-ids> ::= "[" <id-particao> " , " <id-particao> "]" <id-particao> ::= 1*<normalID> </pre>
-------	---

(ex.: "lex:br:federal:lei:2000-12-06;126!art5_par2",
 "lex:br:federal:lei:2000-12-06;126![art6,art10]",
 "lex:br:federal:lei:2000-12-06;126![art6,art10],art12,[art20,art30]",
 "lex:br:federal:lei:1993-07-20;lei.audiovisual!tit1")

A semântica desses fragmentos é fixada pela semântica XPointer (DeRose et al. 2002) no seguinte mapeamento (esboço de algoritmo para mapeamento da sintaxe LexML-fragmento em sintaxe XPointer):

1. todo `<id-particao>` é convertido para `"id('strIdParticao')"`
2. todo `<intervalo-ids>` é convertido para `"id('strId1')/range-to(id('strId2'))"`
3. o conjunto com mais de um fragmento (separado por vírgula) é convertido em um conjunto separado por `" | "` (*pipe*).
4. a *string* resultante é convertida para `"xpointer(strResultante)"`

Aplicando este algoritmo ao fragmento "art5_par2", gera como resultado o valor `"xpointer(id('art5_par2'))"`.

O mapeamento de `"[art6,art10],art12,[art20,art30]"` resulta em `"xpointer(id('art6')/range-to(id('art10'))|id('art12')|id('art20')/range-to(id('art30')))"`.

11.1. Fragmento da Publicação Oficial

O elemento `<fragmento>` permite especificar, no caso da publicação oficial, a página inicial (prefixo "pag") e, se necessário, separado por sublinhado ("_"), a coluna (prefixo "col") onde o documento foi publicado.

(ex.: urn:lex:br:imprensa.nacional:publicacao.oficial;diario.oficial.uniao;secao.1:1997-07-17;222:pag15081_col1).

Caso a publicação oficial possua um identificador único do documento visível na própria publicação, pode-se utilizar o prefixo “doc” seguido do número de identificação do documento.

(ex.: urn:lex:br:superior.tribunal.justica:publicacao.oficial;diario.justica.eletronico;2008-11-05;251;doc4312835).

12. Elemento <versao>

12.1. As Versões e Visões de um Texto

Podem existir diversas versões do texto de um documento, cada uma delas caracterizada pelo intervalo temporal no qual o texto pode ser considerado referenciável ou válido. O período de vigência ou validade de uma versão termina com o início da versão sucessiva.

O conceito de versão está ligado ao evento de entrada em vigor do texto no ordenamento jurídico (no caso de normas) ou de validade (no caso de proposições legislativas e julgados). O conceito de visão, por sua vez, está ligado ao evento que gerou uma nova versão ou uma nova visão de uma versão existente.

Em alguns casos, como na publicação de uma retificação, passam a existir duas visões de uma mesma versão do texto do documento: a visão anterior e a visão posterior ao evento de retificação. Além da retificação, outros eventos, que serão detalhados a seguir, podem gerar novas visões para uma mesma versão.

A data da versão é originada a partir da data do evento:

- o de entrada em vigor de um novo ato normativo;
- o de entrada em vigor das modificações realizadas por outros atos;
- o de republicação por erro, realizada após o *vacatio legis*.
- o de publicação de uma iniciativa de projeto de norma;
- o de consolidação de uma emenda ao projeto de norma;

A data da visão é originada a partir da data do evento:

- o de publicação de um novo ato normativo ou projeto de norma;
- o de publicação de correções devido a erros na publicação (retificações ou erratas);
- o de publicação de decisão judicial com reflexo na norma jurídica;
- o de entrada em vigor das modificações realizadas por outros atos normativos antes da entrada em vigor do ato alterado;
- o de republicação por erro realizada antes do *vacatio legis*.
- o de publicação da derrubada de veto parcial.
- o (qualquer) que gere alguma variante do texto de uma versão.

12.2. Casos Especiais

A criação do texto atualizado de uma norma pode considerar uma ou mais datas de vigência, de acordo com os seguintes tipos:

- multivigente: quando, no interior de um único texto, são adicionadas marcações adequadas para identificar as partes modificadas (inseridas, substituídas e suprimidas) e os relativos períodos de vigência. Um documento desse tipo pode ser capaz, de maneira dinâmica, de se apresentar de formas distintas, de acordo com a data de vigência desejada;
- monovigente: quando, por outro lado, para cada variação sofrida pelo texto a uma certa data, é criado um novo texto distinto. Cada texto é, portanto, caracterizado por um período próprio de validade ou vigência.

12.3. Identificação da Versão

Para identificar, no tempo, as diversas versões de um mesmo documento, o nome uniforme de cada versão é construído agregando-se uma desinência particular àquela do documento original.

Os elementos que qualificam o documento individual são definidos conforme as seguintes regras:

URN Canônica	<pre> <versao> ::= [(<datas> "versao.original" "inicio.vigencia" "multivigente") [";" <evento> [";" <visao>]]] </pre>
URN de Referência	<pre> <versao> ::= [(<datas> "versao.original" "inicio.vigencia" "multivigente" "versao.vigente.em;" <data> "versao.eficaz.em;" <data> "versao.consultada.em;" <data>) [";" <evento> [";" <visao>]]] </pre>
Comum	<pre> <evento> ::= 1*<normal> <visao> ::= <datas> </pre>

A especificação da <versao> inicia com a data do início de período de vigência ou validade do documento. Caso o sistema de informação não possua esta data,

pode-se especificar, no caso de normas, o valor “início.vigência” e, para os demais casos, o valor “versão.original”. No caso do texto multiversão, pode-se especificar uma lista de datas de vigência separadas por “,” (vírgula), um período entre duas datas delimitadas por “[” e “]” (ex.: “[2003-05-01,2005-04-02]”) ou ainda o valor “multivigente”.

No caso das URNs de Referência, é possível referenciar o texto vigente ou eficaz em uma determinada data, ou ainda o texto consultado em uma data específica.

O elemento <evento> indica o evento que define a visão de um documento. Caso não seja informado, considera-se a publicação como o evento default. Um vocabulário específico (ver Parte 6 – Vocabulário Controlado) define os possíveis valores dos eventos.

O elemento <visão> reporta a data do evento que define uma visão do documento. No caso do texto multivisão, pode-se especificar uma lista de datas de visões separadas por “,” (vírgula) ou um período entre duas datas delimitadas por “[” e “]” (ex.: “[2003-05-01,2005-04-02]”).

12.4. Exemplo - CDC

Para exemplificar a relação entre os componentes do elemento visão, vamos analisar como os eventos de publicação, alteração (neste exemplo consideramos apenas as duas primeiras alterações desta norma) e retificação do CDC (Código de Defesa do Consumidor) são modelados. Os eventos citados são apresentados na Tabela 2 e são a base para a construção das URNs de cada versão, apresentadas na Tabela 3.

Tabela 2. Eventos do ciclo de vida do CDC (Lei 8078/1990).

Evento	Nome do Evento	Versão	Visão
E1	Criação da norma com entrada em vigor em 11/03/1991 (180 dias da publicação)	Vigência* 11/03/1991	Assinatura 11/09/1990
E2	Publicação da norma	Vigência 11/03/1991	Publicação 12/09/1990
E3	Entrada em vigor da norma alteradora, Lei 8656/1993 (publicada em 22/05/1993, com entrada em vigor na publicação)	Entrada em vigor da alteração. 22/05/1993	Publicação 22/05/1993
E4	Entrada em vigor da norma alteradora, Lei 8703/1993 (publicada em 08/09/1993, com entrada em vigor na publicação)	Entrada em vigor da alteração. 08/09/1993	Publicação 08/09/1993
E5	Retificação do texto publicado (E2)	[Todas as vigências desde evento E2]	Publicação 10/01/2007

Tabela 3. URNs do CDC (em ordem da ocorrência de eventos)

Ref	Tipo URN	Sufixo da URN (“urn:lex:br:federal:lei” + [sufixo da urn])
E1	Doc. Individual	1990-09-11;8078@1991-03-11;assinatura;1990-09-11
E2	Doc. Individual	1990-09-11;8078@1991-03-11;publicacao;1990-09-12
E3	Doc. Individual	1990-09-11;8078@1993-05-22;alteracao;1993-05-22
E4	Doc. Individual	1990-09-11;8078@1993-09-08;alteracao;1993-09-08
E5	Doc. Individual	1990-09-11;8078@1991-03-11;retificacao;2007-01-10
E5	Doc. Individual	1990-09-11;8078@1993-05-22;retificacao;2007-01-10
E5	Doc. Individual	1990-09-11;8078@1993-09-08;retificacao;2007-01-10

12.5. Valores *Default* para os Elementos da Versão

Por convenção, iremos considerar os seguintes valores *default* para as URNs de documentos individuais que não especifiquem valores para a data da versão, evento e visão:

- Data da Versão (*default*) – A versão do início de vigência (norma) ou validade de um texto (julgados e proposições legislativas), representado respectivamente pelo valores *default* “inicio.vigencia” e “versao.original”.
- Evento (*default*) - O evento da primeira publicação do documento, representado pelo valor *default* “publicacao”.
- Data da Visão (*default*) – A data do evento que tornou a versão pública (normalmente, a data da primeira publicação).

13. Elemento <forma>

O elemento <forma> identifica a forma de expressão do recurso endereçado. Caso a forma seja textual, deve-se informar o código da língua em que o texto está codificado. Nos casos de elementos não textuais, deve-se informar a natureza do recurso de acordo com o vocabulário controlado de tipo de conteúdo.

Comum	<pre> <forma> ::= <tipo-forma> [";" <lingua>] *["," <lingua>]] <tipo-forma> ::= 1*<normal> <lingua> ::= <lingua-codigo> ["-" <lingua-script>] ["-" <lingua-regiao>] <lingua-codigo> ::= 1*<minuscula> <lingua-script> ::= 1*<minuscula> <lingua-regiao> ::= 1*<minuscula> </pre>
-------	---

(ex.: federal:lei:1992-05-11;8421@1992-05-12;assinatura;1992-05-12~texto;pt-br

```
federal:lei:1992-05-11;8421;anexo.1,desenho.bandeira.nacional  
@1992-05-12;assinatura;1992-05-12~imagem )
```

No caso de um documento que possui seu conteúdo expresso em múltiplas línguas, deve-se especificar cada código separado por vírgulas.

(ex.: `federal:lei:1992-05-11;8421@1992-05-12;assinatura;1992-05-12~texto;pt-br,es-ar`)

Por convenção, considera-se o valor “`texto;pt-br`” como *default* para o elemento `<forma>` de uma URN de documento individual.

14. Referências

14.1. Bibliográficas

Moats, R; Sollins, K. R. *URN Syntax*, RFC 2141, Maio 1997.

Berners-Lee, T; Fielding, R; Masinter, L. *Uniform Resource Identifiers (URI): Generic Syntax*, RFC 2396, Agosto 1998.

DeRose, S; Maler, E; Daniel Jr., Ron. *XPointer xpointer() Scheme*. W3C Working Draft, Dezembro de 2002. Disponível em: <http://www.w3.org/TR/xptr-xpointer/>. Acessado em: 4 de Novembro de 2008.

14.2. Sites

<http://cidoc.ics.forth.gr/> - CIDOC CRM

<http://www.ifla.org/VII/s13/frbr/> - IFLA / FRBR

15. Lista de Abreviaturas e Siglas

BNF - Backus-Naur Form

CDC – Código de Defesa do Consumidor

HTML – Hypertext Markup Language

IETF - Internet Engineering Task Force

ITTIG - Istituto di Teoria e Tecniche dell'Informazione Giuridica

NIR – Norme in Rete

ODF – Open Document Format

PDF – Portable Document Format

RDS - Resolver Discovery Service

RFC - Request For Comments

URI - Uniform Resource Identifier

URL - Uniform Resource Locator

URN - Uniform Resource Name

XML – Extensible Markup Language

16. Créditos

16.1. Editores (LexML Brasil)

Fernando Ciciliati (Senado Federal / Interlegis)

João Alberto de Oliveira Lima (Senado Federal / Prodasen)

16.2. Grupo de Trabalho LexML (em Dezembro de 2008)

Alfredo Luiz Campos Júnior (Câmara dos Deputados / CENIN)

Carlos Corrêa Gonçalves (Tribunal Superior Eleitoral)

Cláudio Morale (Senado Federal / Interlegis)

Cláudio Mori (Tribunal Superior Eleitoral)

Claudson dos Santos Melo (Tribunal Superior do Trabalho)

Dalva Luca (Ministério da Justiça)

Fernando Teixeira (Câmara dos Deputados / CENIN)

Flávia Lacerda Oliveira de Macedo (Tribunal de Contas da União)

Flávio Henrique Rocha e Silva (Supremo Tribunal Federal)

Flávio Heringer (Senado Federal)

Manuel de Medeiros Dantas (Advocacia Geral da União)

Jean Rodrigo Ferri (Senado Federal / Interlegis)

João Alberto de Oliveira Lima (Senado Federal / Prodasen)

João Batista de Holanda Neto (Senado Federal / Prodasen)

João R. Kramer Santana (Tribunal de Contas da União)

Jonatas Rodrigues (Procuradoria Geral da República)

Luís Henrique Raja G. Mitchell (Tribunal de Contas da União)

Paulo André Mattos de Carvalho (Tribunal de Contas da União)

Paulo de Tarso Penna da Costa (Superior Tribunal de Justiça)

Paulo Martins Inocêncio (Conselho da Justiça Federal)

Ricardo Bravo (Tribunal de Contas da União)

Sérgio Falcão (Câmara dos Deputados / CENIN)

Virgínia Azevedo (Supremo Tribunal Federal)

16.3. Outras Contribuições

Neide De Sordi (Conselho da Justiça Federal)

Peter de Pádua Krauss (Prefeitura Municipal de São Carlos (SP))

Anexo A – Gramática da URN Canônica

```
*-----
* Estrutura de um nome uniforme (URN) do espaço "lex"
*-----
<URN> ::= "urn:" <NID> ":" <NSS-lex>
<NID> ::= "lex"
*-----
* Estrutura de um nome uniforme do espaço "lex" internacionalizado
*-----
<NSS-lex> ::= <local> ":" <NSS-lex-local>
<local> ::= <codigo-pais> [ <detalhamento-local> ]
*-----
* Estrutura do elemento <local> para o Brasil
*-----
<codigo-pais> ::= "br"
<detalhamento-local> ::=
    ( [ ";" <unidade-federacao> [ ";" <municipio> ] ] )
  | ( [ ";" <unidade-federacao> ] [ ";" <local-judiciario> ] )
<detalhe-ramo> ::= ( ( ("regiao." | "zona.") 1* <numeral> )
    | ( "secao.judiciaria" | "comarca" )
    | ( <unidade-federacao> *[ ", " <unidade-federacao> ] )
    | ( <municipio> *[ ", " <municipio> ] ) ) )
<unidade-federacao> ::= 1*( <minuscula> | "." )
<municipio> ::= 1*( <minuscula> | "." )
<local-judiciario> ::= <ramo-justica> *[ ";" <detalhe-ramo> ]
<ramo-justica> ::= "justica."
    ( "federal" | "trabalho" | "eleitoral" | "militar" | "estadual" )
*-----
* Estrutura de um nome específico do "lex-br"
*-----
<NSS-lex-local> ::= <documentoComplexo>
    | ( <documentoIndividual> [ "!" <fragmento> ] )
<documentoIndividual> ::= <documentoComplexo> [ "@" <versao> ] [ "~" <forma> ]
<documentoComplexo> ::= <documento>
```

```
*-----
* Estrutura do elemento <documento>
*-----
<documento> ::= <autoridade> ":" <tipo-documento> ":" <descriptor>
*-----
* Estrutura do elemento <autoridade>
*-----
<autoridade>      ::= <sujeito> *["," <sujeito> ]
<sujeito>         ::= ( <instituicao> *[";" <orgao> ] [ ";" <funcao> ] )
                    | <cargo>
<instituicao>      ::= 1*<normal>
<orgao>           ::= 1*<normal>
<funcao>          ::= 1*<normal>
<cargo>           ::= 1*<normal>
*-----
* Estrutura do elemento <tipo-documento>
*-----
<tipo-documento> ::= ( ( <tipo-norma>
                        | <tipo-jurisprudencia>
                        | <tipo-projeto-norma>
                        ) [ ";" <nome-subtipo-sequenciador> ] )
                    | ( "publicacao.oficial;" <nome-periodico-oficial>
                        [ ";" <nome-secao-periodico-oficial> ]
                        [ ";" <detalhe-extra-suplemento> ] )
<tipo-projeto-norma> ::= 1*<normal>
<tipo-norma>         ::= 1*<normal>
<tipo-jurisprudencia> ::= 1*<normal>
<nome-subtipo-sequenciador> ::= 1*<normal>
<nome-periodico-oficial>   ::= 1*<normal>
<nome-secao-periodico-oficial> ::= 1*<normal>
<detalhe-extra-suplemento> ::=
    ( "edicao.extra" | "suplemento" ) [ "." 1*<numeral> ]
*-----
* Estrutura do elemento <descriptor>
*-----
```

```
<descriptor> ::= <datasOuAno> ";" <identificadores>
                * [ ";" <id-componente> [ "," <titulo-componente> ] ]
                [ ";"retificacao." <seq-retificacao> ]
<identificadores> ::= ( <id-documento> * [ "," <id-documento> ] )
                    | ( <numero-lex> )
                    | ( <numero-seq> )
<id-documento> ::= 1* <normalID>
<numero-lex> ::= "lex-" 1* <numeral>
<numero-seq> ::= "seq-" <sigla-orgao> "-" 1* <numeral>
<sigla-orgao> ::= 1* <normal>
<id-componente> ::= 1* <normal>
<titulo-componente> ::= 1* <normal>
<seq-retificacao> ::= 1* <numeral>
<datasOuAno> ::= <datas> | <ano>
*-----
* Estrutura do elemento <fragmento>
*-----
<fragmento> ::= ( <id-particao> | <intervalo-ids> ) * [ "," <fragmento> ]
<intervalo-ids> ::= "[" <id-particao> "," <id-particao> "]"
<id-particao> ::= 1* <normalID>
*-----
* Estrutura do elemento <versao>
*-----
<versao> ::= [ ( <datas>
                | "versao.original"
                | "inicio.vigencia"
                | "multivigente" )
                [ ";" <evento> [ ";" <visao> ] ] ]
<evento> ::= 1* <normal>
<visao> ::= <datas>
*-----
* Estrutura do elemento <forma>
*-----
<forma> ::= <tipo-forma> [ ";" <lingua> ] * [ "," <lingua> ] ]
<tipo-forma> ::= 1* <normal>
```

```
<lingua> ::= <lingua-codigo>
           [ "-" <lingua-script> ] [ "-" <lingua-regiao> ]

<lingua-codigo> ::= 1*<minuscula>
<lingua-script> ::= 1*<minuscula>
<lingua-regiao> ::= 1*<minuscula>

*-----
* Estrutura do elemento <datas>
*-----

<datas> ::= ( <data> *["," <data> ] ) | <periodo>
<periodo> ::= "[" <data> "," <data> "]"
<data> ::= <ano> "-" <mes> "-" <dia>
<ano> ::= 4<numeral>
<mes> ::= 2<numeral>
<dia> ::= 2<numeral>

*-----
* Caracteres admitidos
*-----

<aceitos-lex> ::= <normal> | <reservado> | <detalhaID> | <outro>
<normal> ::= <minuscula> | <numeral> | "."
<normalID> ::= <normal> | <detalhaID>
<minuscula> ::= "a" | "b" | "c" | "d" | "e" | "f" | "g" | "h" |
               "i" | "j" | "k" | "l" | "m" | "n" | "o" | "p" | "q" |
               "r" | "s" | "t" | "u" | "v" | "w" | "x" | "y" | "z"
<numeral> ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"
<reservado> ::= ":" | "@" | ";" | "," | "[" | "]" | "!" | "~"
<detalhaID> ::= "_" | "-"
<outro> ::= "=" | "$"
```

Anexo B – Gramática da URN de Referência

```
*-----
* Estrutura de um nome uniforme (URN) do espaço "lex"
*-----
<URN> ::= "urn:" <NID> ":" <NSS-lex>
<NID> ::= "lex"
*-----
* Estrutura de um nome uniforme do espaço "lex" internacionalizado
*-----
<NSS-lex> ::= <local> ":" <NSS-lex-local>
<local> ::= <codigo-pais> [ <detalhamento-local> ]
*-----
* Estrutura do elemento <local> para o Brasil
*-----
<codigo-pais> ::= "br"
<detalhamento-local> ::=
    ( [ ";" <unidade-federacao-ou-uf> [ ";" <municipio> ] ] )
  | ( [ ";" <unidade-federacao-ou-uf> ] [ ";" <local-judiciario> ] )
<detalhe-ramo> ::= ( ( ("regiao." | "zona.") 1* <numeral> )
  | ( "secao.judiciaria" | "comarca" )
  | ( <unidade-federacao-ou-uf> *[ "," <unidade-federacao-ou-uf> ] )
  | ( <municipio> *[ "," <municipio> ] ) )
<unidade-federacao-ou-uf> ::= <unidade-federacao> | 2<minuscula>
<unidade-federacao> ::= 1*( <minuscula> | "." )
<municipio> ::= 1*( <minuscula> | "." )
<local-judiciario> ::= <ramo-justica> *[ ";" <detalhe-ramo> ]
<ramo-justica> ::= "justica."
    ("federal" | "trabalho" | "eleitoral" | "militar" | "estadual")
*-----
* Estrutura de um nome específico do "lex-br"
*-----
<NSS-lex-local> ::= <documentoComplexo>
    | ( <documentoIndividual> [ "!" <fragmento> ] )
<documentoIndividual> ::= <documentoComplexo> [ "@" <versao> ] [ "~" <forma> ]
```

```
<documentoComplexo> ::= <documento>
*-----
* Estrutura do elemento <documento>
*-----
<documento> ::= <autoridade> ":" <tipo-documento> ":" <descriptor>
*-----
* Estrutura do elemento <autoridade>
*-----
<autoridade> ::= <sujeito> *["," <sujeito> ]
<sujeito> ::= ( <instituicao> *[";" <orgao> ] [ ";" <funcao> ] )
| <cargo>
<instituicao> ::= 1*<normal>
<orgao> ::= 1*<normal>
<funcao> ::= 1*<normal>
<cargo> ::= 1*<normal>
*-----
* Estrutura do elemento <tipo-documento>
*-----
<tipo-documento> ::= ( ( <tipo-norma>
| <tipo-jurisprudencia>
| <tipo-projeto-norma>
) [ ";" <nome-subtipo-sequenciador> ] )
| ( "publicacao.oficial;" <nome-periodico-oficial>
[";" <nome-secao-periodico-oficial> ]
[";" <detalhe-extra-suplemento> ] )
<tipo-projeto-norma> ::= 1*<normal>
<tipo-norma> ::= 1*<normal>
<tipo-jurisprudencia> ::= 1*<normal>
<nome-subtipo-sequenciador> ::= 1*<normal>
<nome-periodico-oficial> ::= 1*<normal>
<nome-secao-periodico-oficial> ::= 1*<normal>
<detalhe-extra-suplemento> ::= "edicao.extra" | "suplemento"
*-----
* Estrutura do elemento <descriptor>
*-----
```

```
<descriptor> ::= ( ( <datas> [ ";" <identificadores> ] )
                    | ( <datasOuAno> ";" <identificadores> )
                    | ( [ <datasOuAno> ";" ] <apelido-documento> )
                    )
                *[";" <id-componente> ["," <titulo-componente> ] ]
                [ ";"retificacao." <seq-retificacao> ]

<identificadores> ::= ( <id-documento> *[";" <id-documento> ] )
                    | ( <numero-lex> )
                    | ( <numero-seq> )

<apelido-documento> ::= 1*<normal>
<id-documento>       ::= 1*<normalID>
<numero-lex>        ::= "lex-" 1*<numeral>
<numero-seq>       ::= "seq-" <sigla-orgao> "-" 1*<numeral>
<sigla-orgao>      ::= 1*<normal>
<id-componente>    ::= 1*<normal>
<titulo-componente> ::= 1*<normal>
<seq-retificacao>  ::= 1*<numeral>
<datasOuAno>      ::= <datas> | <ano>

*-----
* Estrutura do elemento <fragmento>
*-----
<fragmento> ::= ( <id-particao> | <intervalo-ids> ) *[";" <fragmento> ]
<intervalo-ids> ::= "[" <id-particao> "," <id-particao> "]"
<id-particao> ::= 1*<normalID>

*-----
* Estrutura do elemento <versao>
*-----
<versao> ::= [ ( <datas>
                | "versao.original"
                | "inicio.vigencia"
                | "multivigente"
                | "versao.vigente.em;" <data>
                | "versao.eficaz.em;" <data>
                | "versao.consultada.em;" <data> )
              [ ";" <evento> [ ";" <visao> ] ] ]
```

```
<evento> ::= 1*<normal>
<visao> ::= <datas>
*-----
* Estrutura do elemento <forma>
*-----
<forma> ::= <tipo-forma> [ ";" <lingua> ] *["," <lingua> ] ]
<tipo-forma> ::= 1*<normal>
<lingua> ::= <lingua-codigo>
                [ "-" <lingua-script> ] [ "-" <lingua-regiao> ]
<lingua-codigo> ::= 1*<minuscula>
<lingua-script> ::= 1*<minuscula>
<lingua-regiao> ::= 1*<minuscula>
*-----
* Estrutura do elemento <datas>
*-----
<datas> ::= ( <data> *["," <data> ] ) | <periodo>
<periodo> ::= "[" <data> "," <data> "]"
<data> ::= <ano> "-" <mes> "-" <dia>
<ano> ::= 4<numeral>
<mes> ::= 2<numeral>
<dia> ::= 2<numeral>
*-----
* Caracteres admitidos
*-----
<aceitos-lex> ::= <normal> | <reservado> | <detalhaID> | <outro>
<normal> ::= <minuscula> | <numeral> | "."
<normalID> ::= <normal> | <detalhaID>
<minuscula> ::= "a" | "b" | "c" | "d" | "e" | "f" | "g" | "h" |
                "i" | "j" | "k" | "l" | "m" | "n" | "o" | "p" | "q" |
                "r" | "s" | "t" | "u" | "v" | "w" | "x" | "y" | "z"
<numeral> ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"
<reservado> ::= ":" | "@" | ";" | "," | "[" | "]" | "!" | "~"
<detalhaID> ::= "_" | "-"
<outro> ::= "=" | "$"
```